Řešení soustav nelineárních rovnic

Řešení soustav nelineárních rovnic užitím rezultantu (eliminantu)

PŘÍKLAD: Určete všechna řešení soustavy rovnic

$$x^2 + 2\,y^2 - 2 = 0,$$
$$x^2 + x\,y + y^2 - 2 = 0.$$

```
> restart;
> f:=x^2+2*y^2-2; g:=x^2+x*y+y^2-2;
```

$$f := x^2 + 2\,y^2 - 2$$
$$g := x^2 + x\,y + y^2 - 2$$

```
> plots[implicitplot]([f,g],x=-10..10,y=-10..10,numpoints=10000,scaling=constrained,color=[red,blue]);
```



1. Klasické řešení

```
> Res:=solve({f,g},{x,y});
```

$$Res := \{y = 0, x = \mathrm{RootOf}(\_Z^2 - 2)\}, \{x = \mathrm{RootOf}(3\,\_Z^2 - 2), y = \mathrm{RootOf}(3\,\_Z^2 - 2)\}$$

```
> Res1:=allvalues(Res[1]); Res2:=allvalues(Res[2]);
```

$$Res1 := \{y = 0, x = \sqrt{2}\}, \{y = 0, x = -\sqrt{2}\}$$
$$Res2 := \{x = \frac{\sqrt{6}}{3}, y = \frac{\sqrt{6}}{3}\}, \{x = -\frac{\sqrt{6}}{3}, y = -\frac{\sqrt{6}}{3}\}$$

```
> P1:=eval([x,y],Res1[1]); P2:=eval([x,y],Res1[2]);
  P3:=eval([x,y],Res2[1]); P4:=eval([x,y],Res2[2]);
```

$$P1 := [\sqrt{2}, 0]$$
$$P2 := [-\sqrt{2}, 0]$$

$$P3 := \left[\frac{\sqrt{6}}{3}, \frac{\sqrt{6}}{3}\right]$$

$$P4 := \left[-\frac{\sqrt{6}}{3}, -\frac{\sqrt{6}}{3}\right]$$

2. Řešení užitím rezultantu

Vytvoříme Sylvesterovu matici polynomů f(x), g(x)

Poznámka: Proměnnou y v tuto chvíli neuvažujeme jako neznámou, potlačíme ji. Proto hovoříme o polynomech f(x), g(x), nikoliv o polynomech f(x,y), g(x,y).

```
> Syl:=LinearAlgebra[SylvesterMatrix](f,g,x);
```

$$Syl := \begin{bmatrix} 1 & 0 & 2y^2-2 & 0 \\ 0 & 1 & 0 & 2y^2-2 \\ 1 & y & y^2-2 & 0 \\ 0 & 1 & y & y^2-2 \end{bmatrix}$$

Determinant Sylvesterovy matice nazýváme rezultant (též eliminant). V tomto případě se jedná o polynom jedné neznámé y. Rezultant položíme roven nule a vypočítáme příslušné hodnoty y

```
> Rezultant:=linalg[det](Syl);
```

$$Rezultant := 3y^4 - 2y^2$$

Pro výpočet rezultantu můžeme použít i funkci "resultant(f,g,x)", která je v Maple implementována:

```
> Rezultant:=resultant(f,g,x);
```

$$Rezultant := 3y^4 - 2y^2$$

```
> Reseni_y:=solve(Rezultant,{y});
```

$$Reseni\_y := \{y=0\}, \{y=0\}, \left\{y=\frac{\sqrt{6}}{3}\right\}, \left\{y=-\frac{\sqrt{6}}{3}\right\}$$

Jednotlivá řešení pro y postupně dosazujeme do původních rovnic. Dostáváme soustavy dvou rovnic o jedné neznámé x. Ty řešíme.

1. $y = 0$

```
> f_x_1:=eval(f,Reseni_y[1]); g_x_1:=eval(g,Reseni_y[1]);
```

$$f\_x\_1 := x^2 - 2$$

$$g\_x\_1 := x^2 - 2$$

```
> solve(f_x_1,{x}); solve(g_x_1,{x});
```

$$\{x=\sqrt{2}\}, \{x=-\sqrt{2}\}$$

$$\{x=\sqrt{2}\}, \{x=-\sqrt{2}\}$$

2. $y=\frac{\sqrt{6}}{3}$

```
> f_x_2:=eval(f,Reseni_y[3]); g_x_2:=eval(g,Reseni_y[3]);
```

$$f\_x\_2 := x^2-\frac{2}{3}$$

$$g\_x\_2 := x^2+\frac{1}{3}x\sqrt{6}-\frac{4}{3}$$

```
> solve(f_x_2,{x}); solve(g_x_2,{x});
```

$$\{x=\frac{\sqrt{6}}{3}\},\ \{x=-\frac{\sqrt{6}}{3}\}$$

$$\{x=\frac{\sqrt{6}}{3}\},\ \{x=-\frac{2\sqrt{6}}{3}\}$$

3. $y=-\frac{\sqrt{6}}{3}$

```
> f_x_3:=eval(f,Reseni_y[4]); g_x_3:=eval(g,Reseni_y[4]);
```

$$f\_x\_3 := x^2-\frac{2}{3}$$

$$g\_x\_3 := x^2-\frac{1}{3}x\sqrt{6}-\frac{4}{3}$$

```
> solve(f_x_3,{x}); solve(f_x_3,{x});
```

$$\{x=\frac{\sqrt{6}}{3}\},\ \{x=-\frac{\sqrt{6}}{3}\}$$

$$\{x=\frac{\sqrt{6}}{3}\},\ \{x=-\frac{\sqrt{6}}{3}\}$$

```
>
```

PŘÍKLAD: Pro které hodnoty parametru k mají polynomy $\mathrm{f}(x)=x^2+k\,x+2$ **a** $\mathrm{g}(x)=x^3-k\,x+2$ **společný nulový bod?**

```
> restart;
> f:=x^2+k*x+2; g:=x^3-k*x+2;
```

$$f := x^2+k\,x+2$$

$$g := x^3-k\,x+2$$

Vytvoříme Sylvesterovu matici polynomů f(x), g(x)

```
> Syl:=LinearAlgebra[SylvesterMatrix](f,g,x);
```

$$Syl := \begin{bmatrix} 1 & k & 2 & 0 & 0 \\ 0 & 1 & k & 2 & 0 \\ 0 & 0 & 1 & k & 2 \\ 1 & 0 & -k & 2 & 0 \\ 0 & 1 & 0 & -k & 2 \end{bmatrix}$$

Determinant Sylvesterovy matice nazýváme rezultant (též eliminant). V tomto případě se jedná o polynom jedné neznámé k. Rezultant položíme roven nule a vypočítáme příslušné hodnoty k

```
> Rezultant:=linalg[det](Syl);
```

$$Rezultant := 12 + 4\,k^2 + 20\,k - 4\,k^3$$

```
> Reseni_k:=solve(Rezultant,{k});
```

$$Reseni\_k := \{k = 3\}, \{k = -1\}, \{k = -1\}$$

Obě řešení pro k postupně dosadíme do původních rovnic. Pokaždé dostaneme dvojici polynomů. U každého z nich určíme nulové body. V prvním případě mají polynomy f, g jeden společný nulový bod, v druhém případě pak dva.

```
> f_1:=eval(f,Reseni_k[1]); g_1:=eval(g,Reseni_k[1]);
```

$$f\_1 := x^2 + 3\,x + 2$$

$$g\_1 := x^3 - 3\,x + 2$$

```
> Res_f_1:=solve(f_1,x); Res_g_1:=solve(g_1,x);
```

$$Res\_f\_1 := -1, -2$$

$$Res\_g\_1 := -2, 1, 1$$

```
> f_2:=eval(f,Reseni_k[2]); g_2:=eval(g,Reseni_k[2]);
```

$$f\_2 := x^2 - x + 2$$

$$g\_2 := x^3 + x + 2$$

```
> Res_f_2:=solve(f_2,x); Res_g_2:=solve(g_2,x);
```

$$Res\_f\_2 := \frac{1}{2} + \frac{1}{2}\,I\sqrt{7}, \frac{1}{2} - \frac{1}{2}\,I\sqrt{7}$$

$$Res\_g\_2 := -1, \frac{1}{2} + \frac{1}{2}\,I\sqrt{7}, \frac{1}{2} - \frac{1}{2}\,I\sqrt{7}$$

PŘÍKLAD: Určete všechna řešení soustavy rovnic

$$x^2\,y^2 - y^4 + 3\,y - 6 = 0,$$

$$x\,y + y^2 - 2\,y - 4 = 0.$$

```
> restart;
> f:=x^2*y^2-y^4+3*y-6=0; g:=x*y+y^2-2*y-4=0;
```

$$f := x^2\,y^2 - y^4 + 3\,y - 6 = 0$$

$$g := x\,y + y^2 - 2\,y - 4 = 0$$

```
> plots[implicitplot]([f,g],x=-10..10,y=-10..10,numpoints=10000,scaling=constrained,color=[red,blue]);
```



Vytvoříme Sylvesterovu matici polynomů f(x), g(x)

Poznámka: Proměnnou y v tuto chvíli neuvažujeme jako neznámou, potlačíme ji. Proto hovoříme o polynomech f(x), g(x), nikoliv o polynomech f(x,y), g(x,y).

```
> Syl:=LinearAlgebra[SylvesterMatrix](lhs(f),lhs(g),x);
```

$$Syl := \begin{bmatrix} y^2 & 0 & -y^4 + 3\,y - 6 \\ y & y^2 - 2\,y - 4 & 0 \\ 0 & y & y^2 - 2\,y - 4 \end{bmatrix}$$

Determinant Sylvesterovy matice nazýváme rezultant (též eliminant). V tomto případě se jedná o polynom jedné neznámé y. Rezultant položíme roven nule a vypočítáme příslušné hodnoty y

```
> Rezultant:=linalg[det](Syl);
```

$$Rezultant := -4\,y^5 - 4\,y^4 + 19\,y^3 + 10\,y^2$$

```
> factor(Rezultant);
```

$$-y^2\,(y-2)\,(2\,y+5)\,(2\,y+1)$$

```
> Reseni_y:=solve(Rezultant,{y});
```

$$Reseni\_y := \{y = 0\}, \{y = 0\}, \{y = 2\}, \{y = \frac{-5}{2}\}, \{y = \frac{-1}{2}\}$$

Jednotlivá řešení pro y postupně dosazujeme do původních rovnic. Dostáváme soustavy dvou rovnic o jedné neznámé x. Ty řešíme. Jak vidíme hned v případě pro y = 0, nemusí mít vždy řešení.

1. $y = 0$

```
> f_x_1:=eval(f,Reseni_y[1]); g_x_1:=eval(g,Reseni_y[1]);
```

$$f\_x\_1 := -6 = 0$$

$$g\_x\_1 := -4 = 0$$

Nemá řešení

2. $y = 2$

```
> f_x_2:=eval(f,Reseni_y[3]); g_x_2:=eval(g,Reseni_y[3]);
```

$$f\_x\_2 := 4\,x^2 - 16 = 0$$

$$g\_x\_2 := 2\,x - 4 = 0$$

```
> Reseni_x_2:=solve({f_x_2,g_x_2},x);
```

$$Reseni\_x\_2 := \{x = 2\}$$

Řešení:

```
> P1:=eval(eval([x,y],Reseni_x_2),Reseni_y[3]);
```

$$P1 := [2, 2]$$

3. $y = -\frac{5}{2}$

```
> f_x_3:=eval(f,Reseni_y[4]); g_x_3:=eval(g,Reseni_y[4]);
```

$$f\_x\_3 := \frac{25\,x^2}{4} - \frac{841}{16} = 0$$

$$g\_x\_3 := -\frac{5\,x}{2} + \frac{29}{4} = 0$$

```
> Reseni_x_3:=solve({f_x_3,g_x_3},x);
```

$$Reseni\_x\_3 := \{x = \frac{29}{10}\}$$

Řešení:

```
> P2:=eval(eval([x,y],Reseni_x_3),Reseni_y[4]);
```

$$P2 := \left[\frac{29}{10}, \frac{-5}{2}\right]$$

4. $y = -\frac{1}{2}$

```
> f_x_4:=eval(f,Reseni_y[5]); g_x_4:=eval(g,Reseni_y[5]);
```

$$f\_x\_4 := \frac{x^2}{4} - \frac{121}{16} = 0$$

$$g\_x\_4 := -\frac{x}{2} - \frac{11}{4} = 0$$

```
> Reseni_x_4:=solve({f_x_4,g_x_4},x);
```

$$Reseni\_x\_4 := \{x = \frac{-11}{2}\}$$

Řešení:

```
> P3:=eval(eval([x,y],Reseni_x_4),Reseni_y[5]);
```

$$P3 := \left[\frac{-11}{2}, \frac{-1}{2}\right]$$

Závěr: Daná soustava nelineárních rovnic má následující tři řešení

```
> P:={P1,P2,P3};
```

$$P := \left\{\left[\frac{-11}{2}, \frac{-1}{2}\right], [2, 2], \left[\frac{29}{10}, \frac{-5}{2}\right]\right\}$$